
Updated March 2 2011 
Thanks, Alissa, for this correction:

On the second (full) paragraph on page 8, 3rd line from the bottom,

"left" should be "right." 
On page 16, your "silly hypothetical example" :

Should our Model 1 -2 Res Log Likelihood be 9.5?  On the next page it

is listed as 19.5, with the difference from 11.5 being 2.  Then in

the next paragraph you mention the chi-square statistic as 1 (as if

10.5 were the value), but then again use 2 in your example of how to

find it.
Updated
Thanks, Jizhi for this message.  I fixed it.
I am reading your manual about "Doing HLM by SAS PROC MIXED". It is really clear and helpful. I got a couple of questions for you. 
On page 15,  
 
Intercept   IDstudent(IDclass) if for the level-3 variance
 
Intercept   IDclass   is for the level-2 variance.
 
I am just wondering whether IDstudent (IDclass) should be the level-2 variance and IDclass should be the level-3 variance. Also for the interpertation, you wrote " The level of student engagment is mostly between student and then next within students."  But based on the estimates, the beep level had the highest value for variacnes, which should be within students level. 
Updated

Thanks, Travis and John, to find an error in Page 11 (There was some language-level confusion about IDclass and IDstudent).
Thanks, Jiahui, to point out this error:

Q1, on page 6 of your "Doing HLM by SAS PROC MIXED" (word version), the SAS code from line 3 looks the same as the model on page 5, I wonder whether X should be added after intercept on the "random" line as below?

Version 2.3 updated 11/30/2006
Corrected an error on page 14 and 15, table and graph that showed the decomposition of variance for student engagement.  Thanks, Suhas, for pointing this out for me.  

I. Advantage of SAS and SAS’s RROC MIXED

If you are familiar with SAS, SAS PROC MIXED is easy to use or as Japanese say:

BEFORE BREAKFAST (A-SA-ME-SHI-MA-E)

which means you can even do it before you eat breakfast.  If you have a column of an outcome variable, just doing the following will get you the result for a non-conditional model:

proc mixed;

model bsmpv01=;

random intercept/ sub= IDSCHOOL ;

run;

The non-conditional model means the model that uses no predictors.  You can add predictors as fixed effects:
proc mixed;

model bsmpv01= SUSHI;

random intercept/ sub= IDSCHOOL ;

run;
You can treat the coefficient of the predictor as a random effect:

proc mixed;

model bsmpv01= SUSHI;

random intercept JOHN / sub= IDSCHOOL ;

run;
You benefit also from SAS’s email support system (support@sas.com). 

II. What is HLM? 
HLM is a model where we can set coefficients to be “randomly varying” instead of just fixed.  To understand the meaning of “randomly varying,” it is easier to first think about what is NOT VARYING.  Linear regression model like OLS is a model whose coefficients are all fixed (=NOT VARYING).  Coefficients are all fixed in a sense that if the intercept is 5 it really is 5 for everyone in the data set.  If the effect of race is 3 then it really is 3 for everyone in the data set.  This is why we call this type of model fixed effect model. 

So now what is “randomly varying”?  What is random effects?  First of all, we can consider coefficients to be varying by groups.  For example, we say for a school A the intercept is 4 but for school B it is 5.  We say for a school A the race effect is 3, but for a school B the race effect is 6.  But we are not yet talking about random effects.  We are only talking about “different effects” at this point.  If it is just different effects, even OLS can deal with it by using a series of dummy variables.

HLM goes further than just estimating “different effects” for different schools.  Let’s focus on the estimation of intercept/intercepts.  HLM estimates an intercept for each separate group unit and estimate reliability of these estimates.  If we have 50 schools in a data set we will have 50 intercept (or we can call them 50 school effects) and 50 reliability statistics.  And HLM obtains a grand average of all these intercepts using reliability (actually, the inverse of reliability) as weight, so intercepts (or school effects) that are more accurately measured influence the results more heavily (I am not using a precise expression, but this is essentially right.  To be more exact I should be talking about “shrinkage to the grand mean”).  This essentially means that you are using information very wisely, not investing on poorly measured scores but investing more on accurate estimates.  The resulting grand intercept is the one that is very conservative (using every source of reliablity).  HLM estimated this measure very carefully using information from each group unit (both intercept and reliability statistic).
Notice that we treated 50 intercepts (= 50 school effects) as randomly varying in this process.  We have 50 estimates of school intercept and assumed that these are randomly and normally distributed, which provided us with a justification to use reliability/precision as a weight (the use of the notion of weight may be a bit loose here.  I should be talking about “shrinkage to the grand mean”).  HLM pulled imprecisely measured intercepts to the center of the scale, so bad measures don’t influence the result much. 
Treating school effects as random effects is like this.  Let’s imagine School A got 50 point on average and School B got 52 points on math scores.  To treat these as fixed effects means that we feel School A is inherently 50 and School B is 52.  To treat these as random effects means that School A and School B (and other schools together) are trying to help you obtain the ultimate grand average score of the nation as if these schools were a randomly chosen sample.  They are all helping you with both average scores and reliability measures.  Some schools don’t have too many students or variance of the scores are just too big (causing measurement errors to be large).  These schools will tell you, “please don’t take me seriously.  Our measures are poor.”  You may say, “but I cannot just punish you like that?”  They will tell you, it is okay, our being poor on precision is just a random phenomenon.  There is no systematic reason why we were poor on reliability, so please go ahead and pull us to the center of the scale.  There are other schools like us but they are high on precision. Take those guys seriously.”
So far we focused only on random intercepts (=school effects).  We can also talk about race effects or gender effects as random effects.  Each school is helping you with their school average race effect and reliability statistics to obtain the grand average race effect.  This is great because a school that has a little racial heterogeneity will give you an estimate with low reliability score, telling you, “just don’t take us seriously.”
Also I might have sounded as if there are fixed effect models like OLS regression or random effect models like HLM.  In reality, there is only a model and there are fixed effects and random effects.  You get to choose at each variable how you want to treat the effects.  If you feel like schools are a random bunch that is helping you to obtain an ultimate average effect, treat that variable as a random effect.  If you don’t have any reasons to do so or if there is no empirical reason to do so (e.g., no variance in school intercepts), then you treat it as a fixed effect.
Also this entire procedure took care of the issue of error correlation.  As a result of setting the intercept to be different, it corrects for the correlated errors that exist in hierarchically structured data (e.g., data sets that we often use for education research).  Also as a result of setting other coefficients to vary randomly by group units, we may be able to be more true to what is going on in the data.
Let’s examine the simplest models, only-intercept model in HLM and in OLS.  Imagine that our outcome measure is math test score.  The goal of only-intercept model is to derive a mean (b0).  Units of analysis are students who are nested within schools.  We use PROC MIXED to do both OLS and HLM.  Yes, it is possible.  The data is a default data set that comes with SAS installation(You can copy/paste these syntaxes to your SAS and run them).


On the left, we run an OLS regression using PROC MIXED.  The model is a non-HLM model just because we are not using a random statement.  A random statement tells SAS which coefficient should be treated as random effects.  Without it all effects are treated as fixed effects (and therefore this entire model is an OLS regression.  On the right we have an HLM model.  Using a random statement, we told SAS to treat intercept as randomly varying (or in other words, to treat state effects as random effects).
	OLS
	HLM

	Y_jk = b0 + error_jk

	Y_jk=b0 + error_jk + error_k
or 

Level1: Y_jk=b0 + error_jk

Level2: b0=g0 + error_k


	proc mixed data=sashelp.Prdsal3 covtest noclprint;

title "Non conditional model";

class state;

model actual =/solution ddfm=kr;
run;


	proc mixed data=sashelp.Prdsal3 covtest noclprint;

title "Non conditional model";

class state;

model actual =/solution ddfm=kr;

random intercept/sub=state;

run;




I hope the following paragraph makes sense.  In HLM we estimate errors at different levels, such as individual and schools.  If I am a seller of furniture in this sample data, my sales average will be constructed by b0, which is a grand mean, and my own error term and my state’s error term.  In OLS, we only got an individual level error.  In HLM, because we remove the influence of the state effect from my individual effect, we have more confidence in saying that my error is independent, which allows me to have more confidence in the statistical evaluation of b0.

Here I added an interval variable as fixed effects.  This is an example of how this model is an HLM but can have a fixed effect in HLM (obviously).
Y_jk=b0 + b1*X + error_jk + error_k

or 

Level1: Y_jk=b0 + b1*X + error_jk

Level2: b0=g00 + error00_k

Level2: b1=g10

proc mixed data=sashelp.Prdsal3 covtest noclprint;

title "One predictor as an fixed effect";

class state;

model actual =predict/solution ddfm=bw;

random intercept/sub=state;

run;
Here I set the coefficient of PREDICT varying randomly by group units (the intercepts, i.e., state effects, were already varying randomly).  We are treating the PREDICT effect as “random effect.”
Y_jk=b0 + b1*X + error_jk + error_k

or 

Level1: Y_jk=b0 + b1*X + error_jk

Level2: b0=g00 + error00_k

Level2: b1=g10 + error10_k

proc mixed data=sashelp.Prdsal3 covtest noclprint;

title "One predictor as an fixed effect";

class state;

model actual =predict/solution ddfm=bw;

random intercept predict/sub=state;

run;
III. Description of data (for practice)
Data name: esm.sas7dbat

http://www.estat.us/sas/esm.zip
We collected data from high school students in math and science classrooms, using ESM (Experience Sampling Method).  We published in two places.

· “Student Engagement in Mathematics and Science (Chapter 6)” In Reclaiming Urban Education: Confronting the Learning Crisis  Kathryn Borman and Associates  SUNY Press.  2005
· Uekawa, K., Borman K. M., Lee, R. (2007)  Student Engagement in America’s Urban High School Mathematics and Science Classrooms: Findings on Social Organization, Race, and Ethnicity.  The Urban Review, 39 (1), 1-106.

Our subjects come from four cities.  In each city we went to two high schools.  In each high school we visited one math teachers and one science teachers.  We observed two classes that each teacher taught for one entire week.  We gave beepers to five randomly selected girls and five randomly selected boys.  We beeped each 10 minute interval of a class, but we arranged such that each student received a beep (vibration) only once in 20 minutes.  To do this, we divided the ten subjects into two groups.  We beeped the first group at the first ten minute point and 30 minute point.  The second group was beeped at the twenty minute point and 40 minute point.  For a class of 50 minute, which is a typical length in most cities, therefore, we collected two observations from each student.  We visited the class everyday for five days, so we have eight observations per student.  There are also 90 minute class that meets only alternate date.  Students are beeped four times in such long classes at 20 minute interval.  Between each beep, researchers recoded what was going on in the class
Data Structure

About 2000 beep points (repeated measures of Student Engagement)

Nested within about 300 students

Nested within 33 classes

· IDs

· IDstudent (level-1) 

· IDstudent (level2).

· IDclass (level-3)

Data matrix looks like:

Date 

IDclass 
IDstudent 
IDbeep 
TEST …

Monday

A
A1

1
0

Monday

A
A1

3
1

Monday 
A 
A2

2
1

Monday

A
A2

4
1

…continues

Variables selected for exercise

Dependent variable: ENGAGEMENT Scale: This composite is made up of the following 8 survey items.  We used Rasch Model to create a composite.  The eight items are:

When you were signaled the first time today, 
SD  D     A   SA 

•I was paying attention……………………….. 
O     O O     O 

•I did not feel like listening…………………… 
O     O O     O

•My motivation level was high……………….. 
O     O O     O 

•I was bored………………………………….. 
O     O O     O

•I was enjoying class………………………… 
O     O O     O

•I was focused more on class than anything else 
O     O O     O

•I wished the class would end soon………… . 
O     O O     O

•I was completely into class ………………… 
O     O O     O

· Precision_weight: weight derived by 1/(error*error) where error is a standard error of engagement scale derived by WINSTEPS.  This weight is adjusted to add up to the number of observations; otherwise, the variance components change proportion to the sum of weights, while other results remain the same.

· IDBeep: categorical variables for the beeps (1st beep, 2nd beep, up to the 4th in a typical 50 minute class; up to 8th in 90 minute classes that are typically block scheduled.)

· DATE: Categorical variables, Monday, Tuesday, Wednesday, Thursday, Friday

· TEST: Students report that what was being taught in class was related to test.  0 or 1.  Time-varying.

· Fixed Characteristics of individuals

· Hisp: categorical variable for being Hispanic

· SUBJECT: mathematics class versus science class

IV. Specifying Models

Intercept-only model (One-way ANOVA with Random Effects)

Any HLM analysis starts from intercept-only model:

Y=intercept + error1 + error2 + error3

where error1 is within-individual error, error2 is between-individual error, error3 is between-class error.

Intercept is a grand mean that we of courses are interested in.  Also we want to know the size of variance for level-1 (within-individual), level-2 (between-individual), and level-3 (between-class)—to get a sense of how these errors are distributed.  


We need to be a bit careful when we get too much used to calling these simply as level-1 variance, level-2 variance, and level-3 variance.  There is a qualitative difference between level-1 variance, which is residual variance and level-2 and level-3 variance, which are parameter variances.  Level-1 variance/Residual variance is specific to individual cases.  Level-2 variances are the variance of level-2 intercepts (i.e., parameters) and level-3 variances are the variance of level-3 intercepts.  

The point of doing HLM is to set these parameters to vary by group units.  So, syntax-wise, the difference between PROC MIXED and PROC REG (for OLS regression) seems PROC MIXED’s use of RANDOM lines.  With them, users specify a) which PARAMETER (e.g., intercept and beta) to vary and b) by what group units (e.g., individuals, schools) they should vary.  RANDOM lines, thus, are the most important lines in PROC MIXED.
Compare 3-level HLM and 2-level HLM
Libname here "C:\";

/*This is three level model*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class IDclass IDstudent;

model engagement= /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);
/*level2*/

random intercept /sub=IDclass ; 

/*level3*/

run;

/*This is two level model*/

/*Note that I simply put * to get rid of one of the random lines*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class IDclass IDstudent;

model engagement=  /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);
/*level2*/

*random intercept /sub=IDclass ; 

/*level3*/

run;
Description of each statement in PROC MIXED

LET’s EXAMINE THIS line by line

Libname here "C:\";

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class IDclass IDstudent;

model engagement= /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);
/*level2*/

random intercept /sub=IDclass ; 

/*level3*/

run;
PROC MIXED statement
proc mixed data=here.esm covtest noclprint;
“covtest” does a test for covariance components (whether variances are significantly larger than zero.).  The reason why you have to request such a simple thing is that COVTEST is not based on chi-square test that one would use for a test of variance.  It uses instead t-test or something that is not really appropriate.  Shockingly, SAS has not corrected this problem for a while.  Anyways, because SAS feels bad about it, it does not want to make it into a default option, which is why you have to request this.  Not many people know this and I myself could not believe this.  So I guess that means that we cannot really believe in the result of COVTEST and must use it with caution.


When there are lots of group units, use NOCLPRINT to suppress the printing of group names.

CLASS statement

class IDclass IDstudent;
or also

class IDclass IDstudent Hisp;

We throw in the variables that we want SAS to treat as categorical variables.  Variables that are characters (e.g., city names) must be on this line (it won’t run otherwise).  Group IDs, such as IDclass in my example data, must be also in these lines; otherwise, it won’t run.  Variables that are numeric but dummy-coded (e.g., black=1 if black;else 0) don’t have to be in this line, but the outputs will look easier if you do.  One thing that is a pain in the neck with CLASS statement is that it chooses a reference category by alphabetical order.  Whatever group in a classification variable that comes the last when alphabetically ordered will be used as a reference group.  We can control this by data manipulation.  For example, if gender=BOY or GIRL, then I tend to create a new variable to make it explicit that I get girl to be a reference group:
If gender=”Boy” then gender2=”(1) Boy”;

If gender=”Girl” then gender2=”(2) Girl”;
I notice people use dummy coded variables (1 or 0) often, but in SAS there is no reason to do that.  I recommend that you use just character variables as they are and treat them as “class variables” by specifying them in a class statement.  But if you want to center dummy variables, then you will have to use 0 or 1 dummies.  We do that often in HLM setup.

I recommend getting used to the use of class statement – for people who are not using it and instead using dummy-coded variables as numeric variables.  Once you get used to the habit of SAS that singles out a reference category based on the alphabetical order of groups (specified in one classification variable), you feel it is a lot easier (than each time dummy coding a character variable).
MODEL statement

model engagement= /solution ddfm=kr;
ddfm=kr specifics the ways in which the degree of freedom is calculated.  This option is relatively new at the time of this writing (2005).  It seems most close to the degree of freedom option used by Bryk, Raudenbush, and Congdon’s HLM program.  Under this option, the degree of freedom is close to the number of cases minus the number of parameters estimated when evaluating the coefficients that are fixed.  When evaluating the coefficients that are treated as random effects, the option KR uses the number of group units minus the number of parameters estimated.  But the numbers are slightly off, so I think some other things are also considered and I will have to read some literature on KR.  I took a SAS class at the SAS institutes.  The instructor says KR is the way to go, which contradicts the advice of Judith Singer in her famous PROC MIXED paper for the use of ddfm=BW, but again KR is added only lately to PROC MIXED.  And it seems KR makes more sense thatn BW.  BW uses the same DF for all coefficients.  Also BW uses different DF for character variables and numeric variables, which I never understood and no one gave me an explanation.  Please let me know if you know anything about this (( kaz_uekawa@yahoo.com).

SOLUTION (or just S) means “print results” on the parameters estimated.  

Random statement

random intercept /sub=IDstudent(IDclass);
/*level2*/

random intercept /sub=IDclass ; 

/*level3*/
This specifics the level at which you want the intercept (or other coefficients) to vary.  “random intercept /sub=IDstudent(IDclass)” means “make the intercept to vary across individuals (that are nested within classes).”  SUB or subject means subjects.  The use of ( ) confirms the fact that IDstudent in nested within IDclass and this is necessary to obtain the correct degree of freedom when IDstudent is not unique.  If every IDstudent is unique, I believe there is no need to use () and the result would be the same.  My colleagues complain that IDstudent(IDclass) feels counterintuitive and IDclass(IDstudent) makes more intuitive appeal.  I agree, but it has to be IDstudent(IDclass) such that the hosting unit must noted in the bracket.

Let’s make it into a language lesson.  “Random intercept /sub=IDstudent(IDclass);”can be translated into English as, “Please estimate the intercept separately for each IDstudent (that, by the way, is nested within IDclass).”  “Random intercept hisp /sub=IDclass;: can be translated into English as: “Please estimate both the intercept and Hispanic effect separately for each IDclass.”

There is one useful feature of random statement when dealing with class variables.  Class variables are the variables that are made of characters.  Race variable is a good example.  In B, R, and K’s HLM software, you need to dummy code character variables like this.

black=0;

If race=1 then black=1;

white=0;

If race=2 then white=1;
Hispanic=0;

If race=3 then Hispanic=1;

You can do the same in SAS, but in SAS we can put a RACE indicator at a class statement.  But how can we request that the character variable to be set random and estimate how each race group’s slope varies across groups?  This would do it (though race variable is not in this practice data set for now).

random intercept race /sub=IDclass  group=race; 



How to Read Results off the SAS Default Output
RUN THIS.  Be sure you have my data set (http://www.estat.us/sas/esm.zip
) in your C directory.  We are modeling the level of student engagement here.  This is an analysis of variance model or intercept-only model.  No predictor is added.

Libname here "C:\";

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class IDclass IDstudent;

model engagement= /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);
/*level2*/

random intercept /sub=IDclass ; 

/*level3*/

run;
YOU GET:

REML (Restricted Maximum Likelihood Estimation Method) is a default option for estimation method.  It seems most recommended.


 


Recall that the test of covariance is not using chi-square test and SAS has to fix this problem sometime soon in the future!

Intercept     IDstudent(IDclass) is for the level-2 variance

Intercept     IDclass     is for the level-3 variance

Residual  is level-1 variance.
Therefore:

	 
	Variance

	Beep level
	31.9932

	Student level
	23.3556

	Course level
	5.3167


Remember in this study we went to the classroom and gave beepers to kids.  Kids are beeped about 8 times during our research visit.  So beep level is level-1 and it is at the repeated measure level.  You can create a pie chart like this to understand it better.  The level of student engagement is mostly within student and then next betwee students.  There is little between-class differences of students’ engagement level.
[image: image2.emf]Decomposition of Variance: Student 

Engagement Level

Beep level

53%

Student 

level

38%

Course 

level

9%


So, looking at this, I say “engagement depends a lot on students.”  Also I say, “Still given the same students, their engagement level varies a lot!”  And I say, “Gee, I thought classes that we visited looked very different, but there is not much variance between the classrooms.”

You can also use the percentages for each level to convey the same information.  Intraclass correlation is 9% for the class level.  Intraclass correlation is calculated like this:

IC= variance of higher level / total variance

where total variance is a sum of all variances estimated.


Because there are three levels, it is confusing what infraclass correlation means.  But if there are only two levels, it is more straight forward.

IC=level 2 variance/ (level1 variance + level2 variance)


You can use these to evaluate the goodness-of-fit across models.  But when you are using restricted maximum likelihood, you can only compare the two models that are different only by the random effects added.  That is pretty restrictive!

To compare models that vary by fixed effects or random effects, you need to use Full Maximum likelihood methods.  You need to request that by changing an option for estimation method.  For example: 
proc mixed data=here.esm covtest noclprint method= ml;
The way to do goodness of fit test is the same as other statistical methods, such as logistic regression models.  You get the differences of both statistic (e.g., -2 Reg Log Likelihood) and number of parameters estimated.  Check these values using a chi-square table.  Report P.  Here is a hypothetical example (I made up the values).  I use -2 Reg Log Likelihood for this (The first value in the Fit Statistics table above):

Model 1: -2 Reg Log Likelihood 9.5; Number of parameters estimated 5

Model 2: -2 Reg Log Likelihood 11.5; Number of parameters estimated 6

Difference in -2 Reg Log Likelihood =>11.5 – 9.5=2

Difference in Number of parameters estimated => 1
Then I go to a chi-square table.  Find a P-value where the chi-square statistic is 1 and DF is 1 and say if it is statistically significant at a certain alpha level (like 95%).  If you feel lazy about finding a chi-square table, Excel let you test it.  The function to enter into an Excel cell is:

=CHIDIST(x, deg_freedom)

This “returns the one-tailed probability of the chi-squared distribution.  So in the above example, I do this:

=CHIDIST(2, 1)

I get a p-value of 

0.157299265
NOTES: If you want to do this calculation in SAS, the function would be this:

X=1-probchi(2, 1);
So the result suggests that the P is not small enough to claim that the difference is statistically significant.
Finally, this is the coefficient table.  This model is an intercept only model, so we have only one line of result.  If we add predictors, the coefficients for them will be listed in this table.  The degree of freedom is very small, corresponding to the number of classes that we visited minus the number of parameters estimated.  Something else must be being considered here as the number 30.2 indicates.

Replicating 4 models described by Bryk and Raudenbush (1992), Chapter 2
Research Q1: What is the effect of race “being Hispanic” on student engagement, controlling for basic covariates?

Random Intercept model where only the intercept is set random.

/*Hispanic effect FIXED*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class idbeep hisp IDclass IDstudent subject;

model engagement=  idbeep hisp test /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);/*level2*/

random intercept /sub=IDclass ;/*level3*/

run;

Research Q2: Given that Hispanic effect is negative, does that effect vary by classrooms?

Random Coefficients Model where the intercept and Hispanic effect are set random.

/*Hispanic effect random*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class idbeep hisp IDclass IDstudent subject;

model engagement=  idbeep hisp test/solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);/*level2*/

random intercept hisp /sub=IDclass ;/*level3*/ 

run;

Research Q3:Now we know that the effect of Hispanics vary by classroom, can we explain that variation by subject matter (mathematics versus science)? 

Slopes-as-Outcomes Model with Cross-level Interaction

A researcher thinks that even after controlling for subject, the variance in the Hispanics slopes remain (note hisp at the second random line).
/*Hispanic effect random and being predicted by class characteristic*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class idbeep hisp IDclass IDstudent subject;

model engagement=  idbeep test math hisp hisp*math /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);/*level2*/

random intercept hisp/sub=IDclass ;/*level3*/

run;
A researcher thinks that the variance of Hispanic slope can be completely explained by subject matter difference (note hisp at the second random line is suppressed).  Actually this is just a plain interaction term, as we know it in OLS.

A Model with Nonrandomly Varying Slopes

/*Hispanic effect fixed and being predicted by class characteristics/
proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class idbeep hisp IDclass IDstudent subject;

model engagement=  idbeep test math hisp hisp*math /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);/*level2*/

random intercept /*hisp*/ /sub=IDclass ;/*level3*/

run;
Cross-level Interaction, Modeling of Slopes 

While HLM software takes level-specific equation as syntax, SAS PROC MIXED accepts just one equation, or I forget the word for it--maybe one common equation or something as opposed to level-specific.  When thinking about cross-level interaction effect, it is easier to think of it using level-specific representation of the model.

Level-specific way (R& B's HLM software uses this syntax):

Level1: Y=b0 + b1*X + Error

Level2: b0=g00 + g01*W + R0

Level3: b1=g10+g11*W + R1

But this is not how SAS PROC MIXED wants you to program.  You need to insert all higher level equations to be inserted into the level-1 equation.  Above equation will be reduced into one equation in this way:

One equation way (SAS PROC MIXED way)

Y=g00 + g01*W + g10*X +g11*W*X + R1*X + E + R0

How do we translate level-specific way (HLM software way) into a one equation version (PROC MIXED way)?  The rule of thumb is this:

Here is an example: 

Level1: Y=b0 + b1*X + Error

Level2: b0=g00 + g01*W + R0

Level3: b1=g10+g11*W + R1

1. Insert higher level equations into the level-1 equation.

Y=b0 + b1*X + Error

Insert level-2 equations into b's

--> Y=[g00 + g01*W + R0]  + [g10+g11*W + R1 ]*X + Error

Take out the brackets

--> Y=g00 + g01*W + R0  + g10*X +g11*W*X + R1*X + Error

Shuffle the terms around, so it is easier to understand.  And identify which part is the structural part of the model and which part is random components of the model:
--> Y=g00 + g01*W + g10*X +g11*W*X  + R1*X + Error + R0

structural component

    random components

Structural part means that it affects every subject in the data.  Random part varies by a person or a group that he/she belongs to.

Now write fixed effect part of the PROC MIXED syntax.  Focus on the structural part of the equation above.  

proc mixed  ;

class ;

model Y= W   X   W*X;

run;

Next write the random component part.  The rule of thumb is that first you look at the random component, " R1*X + E + R0" and notice which variable is written with *R1 or *any higher level error term.  Throw that variable into the random statement.  Also note that “intercept” is in the random statement to represent R0, which is a level-2 error (or the variance of the random intercepts).
proc mixed  ;

class ;

model Y= W   X   W*X;

random intercept X;

run;

Just one more look at R1*X + Error + R0:

Error is a residual (or level-1 error in HLM terminology) and PROC MIXED syntax assumes it exists, so we don't have to do anything about it. 

Again, R0 is a level-2 error (or level-2 intercept), which is why we said "random intercept .."   

R1*X is the random coefficients for X, which is why we said "random ... X."

The most important thing is to notice which variable gets an asterisk next to it with error terms.  In this case, the error term was R1 and it sat right next to X.  This is why we put X in the random statement.

Advantage of PROC MIXED way of writing one equation

When you start writing it in PROC MIXED way, you may begin to feel that HLM is not so complicated.  For example, something that you used to call level-1 variables, level-2 variables, level-3 variables are now just “variables.”  The reason why we differentiate among the variables of different levels in HLM is because with the software you have to prepare data sets such that there are level-1 data sets, level-2 data sets, and level-3 data sets (though the latest version of HLM has an option of just using one data to process).
In HLM terminology, you use to use terms like level-1 error, level-2 error, level-3 error.  They are now just residuals (=level1 error) and random effects.  "Centering" might have felt like a magical concept unique to HLM, but now it feels like something that is not really specific to HLM but something that is awfully useful in any statistical model.  I even use centering for simple OLS regression because it adds a nice meaning to the intercept.
Finally, you realize HLM is just an another regression.  Without a random statement, a mixed model is the same as OLS regression.

 
This is the same as OLS regression (note that I deleted a random statement):

proc mixed data=ABC covtest noclprint;

title "Non conditional model";

class state;

model actual =/solution ddfm=bw;

run;

This is an HLM (just added a random statement).

proc mixed data=ABC covtest noclprint;

title "Non conditional model";

class state;

model actual =/solution ddfm=bw;

random intercept/sub=state;

run;

Group and grand mean centering
In SAS, we have to prepare a data set such that variables are centered already before being processed by PROC MIXED.  In a way this is an advantage over the HLM software.  It reminds you of the fact that centering is not really specific to HLM.

Grand Mean Centering:

proc standard data=JOHN mean=0;

var XXX YYY;

run;

Compare this to the making of Z-scores:

proc standard data=JOHN mean=0 std=1;

var XXX YYY;

run;

Group Mean Centering

proc standard data=JOHN mean=0;

by groupID;

var XXX YYY;

run;

So why do we do centering?  The issue of centering is introduced in HLM class, but I argue centering is not really specific to HLM.  Why with OLS, we don’t center?  In fact, I do centering even with OLS, so intercept has some meaning.  It corresponds to a value of a typical person.  When I center all the variables, intercept will mean the value for a subject who has 0 on all predictors.  This person must be an average person in the sample for having 0 or mean score on all predictors.  For the ease of interpretation of results, I do standardization of continuous variables (both dependent and independent variables) with a mean of zero and standard deviation of 1 (in other words, Z-score).  For dummy variables, I just do centering.
Centering means subtracting the mean of variable from all observations.  If your height is 172 cm and the mean is 172, then your new value after centering will be 0.  In other words, a person with an average height will receive 0.  A person with 5 in this new variable is a person whose height is average value + 5cm.


Doing proc standard or converting values to a Z-score does two things, setting mean to zero and setting standard deviation to one.  Here I focus on centering part (meaning setting mean to zero).  Centering brings a meaning to the intercept.  Usually when we see results of OLS or other regression models, we don’t really care about the intercept because it is just an arbitrary value when a line crosses the Y-axis on a X-Y plot, as we learned in algebra.  

The intercept obtained by centering predictors is also called “adjusted mean.”  The intercept is now a mean of group when everything else is adjusted.  

When we do HLM, we want to have a substantial meaning to the intercept value in this way, so we know what we are modeling.  Why do we want to know what we are modeling especially with HLM, but not OLS?  With HLM, we talk about variance of group specific intercepts, which is a distribution of an intercept value for each group unit.  We may even plot the distribution of that.  When doing so, we feel better if we know that that intercept has some substantive meaning.  For example, if we see group A’s intercept is 5, it means the typical person in group A’s value is 5.  Without centering, we will be saying “group A’s regression line crosses the Y-axis line at 5,” which does not mean much.

Centering Dummy Variables

In HLM, we sometimes center dummy variables.  This feels strange at first.  For example with gender, we code gender=1 if female and 0 if male. Then we center the value by its mean.  The following proc standard would do the group mean centering, for example:

Group Mean Centering

proc standard data=JOHN mean=1 std=1;

by groupID;

var GENDER;

run;

If there are half men and half women,

Male will receive -.5

Female will receive .5 as values.

The reason why we do this is because we want to assign meanings to the intercept.  When we center gender around its mean, the intercept will mean:

The value that is adjusted for gender composition.

What does this mean?

Under Construction (though somewhere on my website you will find an Excel Sheet that explains this).
V. How to use a TYPE= option in a RANDOM statement (to specify the covariance structure) 
Bryk and Raudenbush’s HLM program provides a correlation matrix among random effects.  This is documented on page 19 and 124 of the HLM book (the second edition).  An example from page 19 is easier to understand.  Using an example of student achievement as an outcome, they talk about the correlation between random intercepts (=school effects) and the SES effect (estimated as random effects).  When correlation is too high (two things are telling same the story redundantly), they recommend (see page 124) that one of the two has to be fixed (based on research purpose and theoretical importance).

How can we get this correlation from SAS PROC MIXED?  To be more exact, when we have to more than one random effects in the model, how do we get a correlation matrix for them?  This is a matter of using a TYPE= option on a random statement.  Compare the two statements below.
UNDER CONSTRUCTION

VI. Doing Econometrics using SAS PROC MIXED’s REPEAT STATEMENT (and TYPE=)
Imagine you collect data from the same individuals many times.  You collect information at an equal interval and you feel your observations/information are correlated in a systematic pattern.  For example, people’s first observation and second observation must be strongly dependent (alike, similar, influenced, etc.).  People’s first and the third observation may not be as highly correlated, but still there must be some correlation there.  All these are violations of independent assumption and in SAS PROC MIXED you use a REPEAT statement to fix it or remove influence of error correlation by modeling something called “covariance matrix.”  The choice of the covariance structure is important, but it is not an ultimate concern of analysts.  But without worrying about it, your model won’t be precise as it obviously violates independent assumptions of errors.  

Below I describe different covariance models.  Choosing the best one is not such a difficult task (despite I spent most of my writing below for that purpose), but I think a real challenge is to achieve a conversion when your model gets too complicated.  For example, if you want to use a REPEAT statement and a RANDOM statement together, the model gets too complicated/demanding.  You may not get clean results.

AR(1) model 

You use AR(1) model when you believe that:

a) there are correlation among information

b) information from time 1 and time 2 (or any other time points that are right next to each other) are most strongly correlated

c) the further the time/distance, the weaker the correlation (and the way this gets weaker is systematic such that ... (I will explain this later)

If you believe your data behaves this way (theory) and if your data shows such a pattern (empirical support), you can use AR(1) model to take care of time-dependent correlation (which means that despite time-dependence among your observations your statistical results will be okay.)

Example.

Imagine I took a statistics class and measured my level of engagement every ten minutes.

Time

Engagement level

Time1

X1

Time2

X2

Time3

X3

Time4

X4

TIme5 
X5

AR(1) model states that  there is the strongest correlation between:

X1 and X2

X2 and X3

X3 and X4

X4 and X5

and the model assumes that it is a fixed correlation estimate (often noted as P)

What about correlation between X1 and X3 or X2 and X4?  AR(1) model says that the correlation between pairs here are P*P (correlation coefficient squared). What about correlation between X1 and X4 or X2 and X5?  AR(1) model says that 

the correlation between pairs here are P*P*P.   Correlation gets weaker and weaker in a systematic way.

There are also other models:

UN

Unstructured Covariance

CS 

Compount Symmetry

TOEP

Toeplitz

ANTE(1)
First-Order Ante Dependent ANTE(1)

(See Chapter 4 of Mixed Models Analyses Using the SAS System)


The goal is to choose the best one that captures reality (of how errors are correlated), so that the model will be right and statistical decisions will be also right.  However, note that it is possible that results won’t be that different after careful examination of different options.  This happens a lot in reality, but it is good to be right (and it keeps us employed.)

Le’t try it.

I use a data set from Chapter 3 of SAS System for Mixed Models (which is an excellent resource).

This data came from:

http://ftp.sas.com/samples/A55235

/*---Data Set 3.2(a)---*/

data weights;

   input subj program$ s1 s2 s3 s4 s5 s6 s7;

   datalines;

 1      CONT      85    85    86    85    87    86    87

 2      CONT      80    79    79    78    78    79    78

 3      CONT      78    77    77    77    76    76    77

 4      CONT      84    84    85    84    83    84    85

 5      CONT      80    81    80    80    79    79    80

 6      CONT      76    78    77    78    78    77    74

 7      CONT      79    79    80    79    80    79    81

 8      CONT      76    76    76    75    75    74    74

 9      CONT      77    78    78    80    80    81    80

10      CONT      79    79    79    79    77    78    79

11      CONT      81    81    80    80    80    81    82

12      CONT      77    76    77    78    77    77    77

13      CONT      82    83    83    83    84    83    83

14      CONT      84    84    83    82    81    79    78

15      CONT      79    81    81    82    82    82    80

16      CONT      79    79    78    77    77    78    78

17      CONT      83    82    83    85    84    83    82

18      CONT      78    78    79    79    78    77    77

19      CONT      80    80    79    79    80    80    80

20      CONT      78    79    80    81    80    79    80

 1      RI        79    79    79    80    80    78    80

 2      RI        83    83    85    85    86    87    87

 3      RI        81    83    82    82    83    83    82

 4      RI        81    81    81    82    82    83    81

 5      RI        80    81    82    82    82    84    86

 6      RI        76    76    76    76    76    76    75

 7      RI        81    84    83    83    85    85    85

 8      RI        77    78    79    79    81    82    81

 9      RI        84    85    87    89    88    85    86

10      RI        74    75    78    78    79    78    78

11      RI        76    77    77    77    77    76    76

12      RI        84    84    86    85    86    86    86

13      RI        79    80    79    80    80    82    82

14      RI        78    78    77    76    75    75    76

15      RI        78    80    77    77    75    75    75

16      RI        84    85    85    85    85    83    82

 1      WI        84    85    84    83    83    83    84

 2      WI        74    75    75    76    75    76    76

 3      WI        83    84    82    81    83    83    82

 4      WI        86    87    87    87    87    87    86

 5      WI        82    83    84    85    84    85    86

 6      WI        79    80    79    79    80    79    80

 7      WI        79    79    79    81    81    83    83

 8      WI        87    89    91    90    91    92    92

 9      WI        81    81    81    82    82    83    83

10      WI        82    82    82    84    86    85    87

11      WI        79    79    80    81    81    81    81

12      WI        79    80    81    82    83    82    82

13      WI        83    84    84    84    84    83    83

14      WI        81    81    82    84    83    82    85

15      WI        78    78    79    79    78    79    79

16      WI        83    82    82    84    84    83    84

17      WI        80    79    79    81    80    80    80

18      WI        80    82    82    82    81    81    81

19      WI        85    86    87    86    86    86    86

20      WI        77    78    80    81    82    82    82

21      WI        80    81    80    81    81    82    83

;

/*---Data Set 3.2(b)---*/

data weight2; 

   set weights;

   time=1; strength=s1; output;

   time=2; strength=s2; output;

   time=3; strength=s3; output;

   time=4; strength=s4; output;

   time=5; strength=s5; output;

   time=6; strength=s6; output;

   time=7; strength=s7; output;

   keep subj program time strength;

run;

proc sort data=weight2; 

   by program time;

run;

SAS SYNTAX

This model below (using a data set from above) is an AR(1) model.

proc mixed data=weight2;

class program subj time;

model strength=program time program*time;

repeated / type=ar(1) sub=subj(program) r rcorr;

run;

The TYPE= part can change, depending on your choice of covariance structure.  The choices are:

Type=UN
Unstructured covariance

Type=TOEP
Toeplitz

type=CS
Compound Symmetry

Challenge of using both repeated and random statements

This model adds a random statement to estimate subject effects. My reference books warn that one needs caution when using both a REPEAT and RANDOM statements because this may be asking the data too much.  For example, if the random statement doesn’t pick up enough variance, it is likely that the model does not produce results.  In one application Raudenbush and Bryk mentions in the HLM book, they “found it impossible to simultaneously estimate the autocorrelation parameter while also allowing randomly varying slopes ... (p. 192).”

proc mixed data=weight2 covtest;

class program subj time;

model strength=program time program*time;

repeated / type=ar(1) sub=subj(program) r rcorr;

random intercept/ sub=subj(program);

run;
Obviously, many of us are both interested in (a) dealing with time-dependent errors and (b) hierarchically structured data.  If we collect data from students, they give us not only time-correlated data (repeated measures), but they are also coming from schools (hierarchically structured data).  

In my references and practices I encountered cases where the AR(1) model converge with both REPEAT and RANDOM statements.  This seems due to a lack of variance (to be estimated by a random statement).  Also note that the simultaneous use of the two statements is technically possible with AR(1) and UN (unstructured covariance).  However, according to my reference books, it does not work with TOEP and CS as it creates “a confounding problem.”  Please let me know if you know any theoretical explanation of this issue (email kaz_uekawa AT yahoo.com).

How to choose the best covariance structure (SAS BOOK)
Read Chapter 4 of Mixed Model Analysis by SAS publication.  It tells you to:

a) Use graphical tools to examine the patterns of correlation over time.

b) Use information criteria to compare relative fit of various covariance structures.

c) Use likelihood ratio tests to compare nested variance structures.

(1) Use graphical tools to examine the patterns of correlation over time.

The SAS book has a great advice on page 230, regarding how to visualize the correlation structure (a, above).  They say you can get a correlation matrix, using an option R and RCORR on the repeated line.  

proc mixed data=weight2 covtest;

class program subj time;

model strength=program time program*time;

repeated / type=un sub=subj(program) r rcorr;

run;
To save this in a SAS data set, you can use an ODS statement like this:

proc mixed data=weight2 covtest;

class program subj time;

model strength=program time program*time;

repeated / type=un sub=subj(program) r rcorr;

ods output  RCorr=kaz_corr;

run;
Do you know how to see a result data set?  Go to a working folder and click on KAZ_CORR.  You can left click it to find an option VIEW IN EXCEL.

This is what I see when I open this data set in Excel:

	Index
	Row
	Col1
	Col2
	Col3
	Col4
	Col5
	Col6
	Col7

	1
	1
	1
	0.9602
	0.9246
	0.8716
	0.8421
	0.8091
	0.7968

	1
	2
	0.9602
	1
	0.9396
	0.877
	0.8596
	0.8273
	0.7917

	1
	3
	0.9246
	0.9396
	1
	0.9556
	0.9372
	0.8975
	0.8755

	1
	4
	0.8716
	0.877
	0.9556
	1
	0.9601
	0.9094
	0.8874

	1
	5
	0.8421
	0.8596
	0.9372
	0.9601
	1
	0.9514
	0.9165

	1
	6
	0.8091
	0.8273
	0.8975
	0.9094
	0.9514
	1
	0.9531

	1
	7
	0.7968
	0.7917
	0.8755
	0.8874
	0.9165
	0.9531
	1


You can inspect this table to decide which covariance structure best fits the empirical reality.

Raudenbush and Bryk’s suggestion

A similar but slightly different approach (conceptually identical?) is mentioned on page 194 of Raudenbush and Bryk’s HLM (Second Edition).  You need to create a covariance data sets as in the following syntax below.  Notice that we are getting a covariance table (R=kaz_R_UN).  Also notice that this is an UN model.  The UN model serves as the most basic model against which other models will be compared to.  The idea is to compare the basic model against other “fancy model.”  You choose whatever fancy model that best replicates the results from the basic model.  They warn that “This is possible only when the number of time points is sufficiently small to allow stable estimation of the unrestricted model (p. 194).”

proc mixed data=weight2 covtest;

class program subj time;

model strength=program time program*time;

repeated / type=un sub=subj(program) r rcorr;

ods output R=kaz_R_UN;

run;
And repeat this for other options, such as CS, AR(1), and TOEP.  Pick whatever replicates the pattern by UN.

proc mixed data=weight2 covtest;

class program subj time;

model strength=program time program*time;

repeated / type=CS sub=subj(program) r rcorr;

ods output  R=kaz_R_CS;

run;
proc mixed data=weight2 covtest;

class program subj time;

model strength=program time program*time;

repeated / type=TOEP sub=subj(program) r rcorr;

ods output  R=kaz_R_TOEP;

run;
proc mixed data=weight2 covtest;

class program subj time;

model strength=program time program*time;

repeated / type=AR(1) sub=subj(program) r rcorr;

ods output  R=kaz_R_AR1;

run;
And then compare these SAS data sets (produced in a working folder):

kaz_R_UN  

kaz_R_CS

kaz_R_TOEP

kaz_R_CS

(b) Use information criteria to compare relative fit of various covariance structures.

Compare AIC, AICC, and BIC to compare the fit of various models (e.g., UN, AR(1), CS, and TOEP). In SAS print-out, highlighted lines would be the section to look at.

UN

                                        Fit Statistics

                             -2 Res Log Likelihood          1234.9

                             AIC (smaller is better)        1290.9

                             AICC (smaller is better)       1295.5

                             BIC (smaller is better)        1348.1
AR(1)

                                        Fit Statistics

                             -2 Res Log Likelihood          1266.8

                             AIC (smaller is better)        1270.8

                             AICC (smaller is better)       1270.8

                             BIC (smaller is better)        1274.9
TOEP

                                        Fit Statistics

                             -2 Res Log Likelihood          1261.1

                             AIC (smaller is better)        1275.1

                             AICC (smaller is better)       1275.4

                             BIC (smaller is better)        1289.4
(c) Use likelihood ratio tests to compare nested variance structures.
For each of UN, AR(1), and TOEP models, find a fit statistics in the first line of the same table.

UN

                                        Fit Statistics

                             -2 Res Log Likelihood          1234.9
                             AIC (smaller is better)        1290.9

                             AICC (smaller is better)       1295.5

                             BIC (smaller is better)        1348.1
AR(1)

                                        Fit Statistics

                             -2 Res Log Likelihood          1266.8
                             AIC (smaller is better)        1270.8

                             AICC (smaller is better)       1270.8

                             BIC (smaller is better)        1274.9
TOEP

                                        Fit Statistics

                             -2 Res Log Likelihood          1261.1
                             AIC (smaller is better)        1275.1

                             AICC (smaller is better)       1275.4

                             BIC (smaller is better)        1289.4
Degree of Freedom will be found in the first line of the dimension table.

UN

                                          Dimensions

                              Covariance Parameters            28
                              Columns in X                     32

                              Columns in Z                      0

                              Subjects                         57

                              Max Obs Per Subject               7
AR(1)

                                          Dimensions

                              Covariance Parameters             2
                              Columns in X                     32

                              Columns in Z                      0

                              Subjects                         57

                              Max Obs Per Subject               7
TOEP

                                          Dimensions

                              Covariance Parameters             7
                              Columns in X                     32

                              Columns in Z                      0

                              Subjects                         57

                              Max Obs Per Subject               7
Collect these two pieces of information from AR(1), TOEP, and UN. 

	 
	 -2 Res Log Likelihood
	DF (=Covariance Parameters)

	UN
	1234.9
	28

	AR(1)
	1266.8
	2

	TOEP
	 1261.1
	 7


You can conduct fit comparison tests of the following combinations:

TOEP vs.  UN

AR1 vs. UN

AR1 vs. TOEP

One is always a reduced model than the other.  To put it differently, one is a special case of the other structure (It is like detergent is a special case of soap).  Let’s pick an example of AR(1) versus UN.  UN estimates 28 covariance parameters, which is a lot of work, while AR(1) estimates only 2.  So UN is working harder at the data set than AR(1).  So how much improvement did UN made over AR(1)?  UN got 31.9 (see table 2 below).  The question is, “was it worth it?”  The UN guy worked so hard and got 31.9, which turned out to be not significant (see Table 4).  In this case, you say, “No, it wasn’t worth the effort.”  You choose AR(1) over UN.


This is a similar to choosing a house to buy.  Imagine a house UN versus a house AR(1).  The UN house costs you a million dollar and the level of pleasure it gives you is 50 (whatever it means).  The AR(1) house costs only a $50,000 and it gets you a pleasure level of 49.  This means that you will waste your money if you choose the UN house.

To do these tests, I created an Excel table.  Just change values in the yellow cells.  The tests will be conducted automatically.  I placed this Excel sheet at:

http://www.estat.us/sas/likelihood_test.xls
	Table 1. Pick these values from SAS outputs

	 
	 -2 Res Log Likelihood
	DF (=Covariance Parameters)

	UN
	1234.9
	28

	AR(1)
	1266.8
	2

	TOEP
	1261.1
	7

	
	
	

	Table 2. Calculate the differences in fit-statistics

	 
	 -2 Res Log Likelihood
	

	TOEP - UN
	26.2
	

	AR1-UN
	31.9
	

	AR1-TOEP
	5.7
	

	
	
	

	
	
	

	Table 3. Calculate the differences in Degree of Freedom

	 
	DF
	

	UN-TOEP
	21
	

	UN-AR1
	26
	

	TOEP-AR1
	5
	

	
	
	

	Table 4. Likelihood Ratio Test
	

	 
	 
	

	TOEP versus UN
	19.89%
	

	AR1 versus UN
	19.64%
	

	AR1 versus TOEP
	33.65%
	


Table 4 says that none of the models that have more number of degree of freedoms (doing a lot of work!) are not improving the fit by working so hard at the data.  This indicates that simpler models are better than elaborate models.

In case you want to replicate this table yourself, here is a version that Excel shows functions.

	Table 1. Pick these values from SAS outputs

	 
	 -2 Res Log Likelihood
	DF (=Covariance Parameters)

	UN
	1234.9
	28

	AR(1)
	1266.8
	2

	TOEP
	1261.1
	7

	
	
	

	Table 2. Calculate the differences in fit-statistics

	 
	 -2 Res Log Likelihood
	

	TOEP - UN
	 =B5-B3
	

	AR1-UN
	 =B4-B3
	

	AR1-TOEP
	 =B4-B5
	

	
	
	

	
	
	

	Table 3. Calculate the differences in Degree of Freedom

	 
	DF
	

	UN-TOEP
	 =$C$3-$C$5
	

	UN-AR1
	 =$C$3-$C$4
	

	TOEP-AR1
	 =$C$5-$C$4
	

	
	
	

	Table 4. Likelihood Ratio Test
	

	 
	 
	

	TOEP versus UN
	 =CHIDIST(B9,B16)

	AR1 versus UN
	 =CHIDIST(B10,B17)

	AR1 versus TOEP
	 =CHIDIST(B11,B18)
	


The use of Weight
PROC MIXED has a weight statement.  But some people tell me that we cannot use a sample weight for the weight statement. Still PROC MIXED’s weight statement must have the same mechanism as PROC MEANS’ weight statement.  I tried the following two syntaxes and they produce the same results.  This means that PROC MIXED uses the same weighting scheme that we are familiar from STAT 101 type of courses.
proc means data=sashelp.class;

weight weight;

var height;run;

proc mixed data=sashelp.class covtest;

weight weight;

model height=/solution;

run;
Please email me if you know more about this subject (kaz_uekawa AT yahoo.com).

Level-specific weight

Bryk and Raudenbush’s HLM program allows users to specify weights that are specific to levels.  In the context of education data analysis, this often means that you use student-weight and school-weight.  SAS PROC MIXED only allows root-level weight, which means that it would allow you to use only student-weight.  When doing a secondary data analysis (with NELS and TIMSS, for example), analysts must decide which approach he/she wants to take.  I have never seen any discussion on this issue, so please email me if you know anything about this topic  (kaz_uekawa AT yahoo.com).
Residual analysis using ODS

ODS (output delivery system) and other SAS procedures allow a researcher to check residuals through a routine procedures.  In the beginning I wasn’t sure what “residual analysis” meant, but I found out that it essentially means that we investigate the random effects that we estimated.  For example, in our model of student engagement, we estimated student level engagement level and class level engagement level.  So the actual values for each student and each teacher’s class are there somewhere behind PROC MXIED.  Because we estimated them, we can of course print them out, graphically examine them, and understand their statistical properties.
In this study from which the data set came, we were actually in the classrooms, observing what was going on.  With procedures above, we were able to compare the distribution of class and student means and our field notes and memories.  Sometimes, we were surprised to see what we thought was a highly engaged classroom was in fact low on engagement level.  Triangulation of quantitative and qualitative data is easy for PROC MIXED has the ability to quickly show a researcher how to check residuals in this way.
Note that I put “solution” at the second random line (it does not matter which random line you put this) and requested that all the data related to that line gets produced and reported.  AND I am requesting this time solutionR to save this information, using ODS.
ODS listing close;/*suppress printing*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class idbeep hisp IDclass IDstudent subject;

model engagement=  /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);
/*level2*/

random intercept /sub=IDclass solution; 
/*level3*/

ods output solutionF=mysolutiontable1

CovParms=mycovtable1 Dimensions=mydimtable1 solutionR=mysolRtable1; 

run;

ODS listing;/*allow printing for the procedures to follow after this*/

I put “ODS listing close” to suppress printing (in SAS’s window); otherwise, we get long outputs.  Don’t forget to put “ODS listing” at the end, so whatever procedures follow it will gain the capacity to print.  

The table produced by solutionR= contains actual parameters (intercepts in this case) for all students and classes.  Unfortunately this table (here arbitrarily named “mysolRtable1”) is messy because it has both individual-mean and class-mean of the outcome.  I selected part of the table for demonstration purpose:
                                                            StdErr

                Obs     Effect      IDclass    IDstudent    Estimate        Pred      DF     tValue     Probt

                  1    Intercept     1_FF       01_1_FF       6.5054      2.3477    1964       2.77    0.0056

                  2    Intercept     1_FF       10_1_FF      -2.7622      2.1226    1964      -1.30    0.1933
                 14    Intercept     .          .                  0      4.8588    1964       0.00    1.0000

                 15    Intercept     .          .                  0      4.8588    1964       0.00    1.0000

                 16    Intercept     .          .                  0      4.8588    1964       0.00    1.0000

                 17    Intercept     1_FF                    -0.2199      1.4898    1964      -0.15    0.8827



and a lot more lines

Estimated Level-2 intercepts (intercepts for each individual or “individual mean” in this case) are the cases where level-2 unit indicator (IDstudent) has the values (e.g., 01_1_FF and 10_1_FF, i.e., observation 1 and 2).  There are lots of them obviously for there are about 300 students to account for.  There are also level-3 intercepts (intercepts for each class, or class mean in this case), which in the above example is observation 17.  You can tell this because the level-2 ID, i.e., IDstudent is empty.  Observation 14,15,and 16 above have dots for both IDs, which (I think) means that they are missing case and were not used in the run.
Perhaps the first thing that we want to do after running the program is to check the distribution of parameters that we decided to set random.  We need to look at it to see if things make sense.

To look at level-3 intercepts (classroom means), select cases when level-2 ID (IDstudent) is empty.

proc print data=mysolRtable1;

where  idstudent="";

run;

/*for something more understandable*/

proc sort data=mysolRtable1 out=sorteddata;

by estimate;

run;

proc timeplot data=sorteddata;

where  idstudent="";

plot  estimate /overlay hiloc npp ref=.5;

id IDclass estimate;

run;
In order to look at level-2 intercepts (student means) and the distribution of values, you must select cases when level-2 ID (IDstudent) is neither empty nor missing.

proc print data=mysolRtable1;

where  idstudent ne "" and idstudent ne ".";

run;

proc univariate data=mysolRtable1 plot;

where  idstudent ne "" and idstudent ne ".";

var estimate;run;

VII. PROC MIXED used along with other SAS functionalities 

· With ODS (Output Delivery System)

ODS is the most powerful feature of SAS, which I think does not exist in other software packages.  Further details of ODS can be found in one of the other manuals that I wrote (downloadable from the same location).  You are able to save results of any SAS procedures, including PROC MIXED, as data sets.

/*Intercept-only model*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class idbeep hisp IDclass IDstudent subject;

model engagement=  /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);
/*level2*/

random intercept /sub=IDclass ; 



/*level3*/

ods output solutionF=mysolutiontable1  CovParms=mycovtable1 ;

run;

/*ADD predictors FIXED*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class idbeep hisp IDclass IDstudent subject;

model engagement=  idbeep hisp test /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);/*level2*/

random intercept /sub=IDclass ;/*level3*/

ods output solutionF=mysolutiontable2  CovParms=mycovtable2;

run;
 “ODS output” tells the program to produce data sets which contain statistical results.  We want to assign names to those data sets.

SolutionF=you_choose_name ( kicks out data with parameter estimates

CovParms=you_choose_name (kicks out data with information about variance

and so on.

Are you curious what you just created?  Do proc print to look at them.

proc print data=type_the_name_of_the_data_here;run;
Forget default outputs.  This is a simple sample of how to create a readable table using ODS.  For an example of how this can be elaborated further, see appendix III, Sample SAS program.
data newtable;set mysolutiontable1 mysolutiontable2;

if probt < 0.1 then sign='+';

if probt < 0.05 then sign='*';

if probt < 0.01 then sign='**';

if probt < 0.001 then sign='***';

if probt < -999 then sign="   ";

run;

/*table a look*/

proc print data=newtable;

var effect estimate sign;

run;

/*create an excel sheet and save it in your C drive*/

PROC EXPORT DATA= newtable  OUTFILE= "C:\newtable.xls" DBMS=EXCEL2000 REPLACE;RUN;

· Wth ODS, and GRAPH

We use the data mycovtable1 obtained in the previous page or the program below.

ODS listing close;/*suppress printing*/

proc mixed data=here.esm covtest noclprint;

weight precision_weight;

class idbeep hisp IDclass IDstudent subject;

model engagement=  /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);
/*level2*/

random intercept /sub=IDclass solution; 

/*level3*/

ods output solutionF=mysolutiontable1  CovParms=mycovtable1 Dimensions=mydimtable1 solutionR=mysolRtable1; 

run;

ODS listing;/*allow printing for the procedures to follow after this*/
Now that the table for variance components (I named mycovtable1) is created, I want to manipulate it, so it is ready for graphing.  Check what is in “mycovtable1” by doing:

proc print data=mycovtable1;run;
It has lots of variables and values that are not really ready for graphing.  In the first data step below, I am modifying the data to look like this, so it is ready to be used by PROC GCHART.

Obs          level           Estimate

 1     Between-Individual     23.6080

 2     Between-Class           5.2838

 3     Within-Individual      31.6213
data forgraphdata;set mycovtable1;

/*crearing a new variable called LEVEL*/

length level $ 18;

if CovParm="Residual" then level="Within-Individual";

if Subject="IDstudent(IDclass)" then level="Between-Individual";

if Subject="IDclass" then level="Between-Class";

goptions cback=black htext=2 ftext=duplex;title1 height=3 c=green 

'Distribution of Variance';

footnote1 h=2 f=duplex c=yellow 'Engagement study: Uekawa, Borman, and Lee';

proc gchart data=forgraphdata;

  pie level /

     sumvar=estimate

     fill=solid

     matchcolor;

run;

/*

Borrowed program code for pie chart from http://www.usc.edu/isd/doc/statistics/sas/graphexamples/piechart1.shtml*/
Other Useful SAS functionalities for HLM analyses
· Creating group-level mean variables

One could use proc means to derive group-level means.  I don’t recommend this since it involves extra steps of merging the mean data back to the main data set.  Extra steps always create rooms for errors.  PROC SQL does it at once.  

proc sql;

create table newdata as

select *, 

mean(variable1) as mean_var1,

mean(variable2) as mean_var2,

mean(variable3) as mean_var3

from here.esm

group by IDCLASS;

Normalizing sample weight right before PROC MIXED

This is a simple example.  F1pn1wt is the name of weight in this data.

proc sql;

create table newdata as

select *,

f1pnlwt * (count(f1pnlwt)/Sum(f1pnlwt)) as F1weight
from here.all; /*proc SQL does not require run statement*/

Sometimes you may want to create more than one weight specific to the dependent variable of your choice—when depending on the outcome, the number of cases are very different.

/*create 1 or 0 marker for present/missing value distinction*/

%macro flag (smoke=);

flag&smoke=1;if &smoke =. then flag&smoke=0; 

%mend flag;

%flag (smoke=F1smoke);

%flag (smoke=F1drink);

%flag (smoke=F1cutclass);

%flag (smoke=F1problem);

%flag (smoke=F1marijuana);

run;

/*create weight specific to each outcome variable.  Note the variables created above are now instrumental in deriving weights*/

proc sql;

create table final as

select *,

f1pnlwt * (count(f1pnlwt)/Sum(f1pnlwt)) as F1weight,

f1pnlwt * (count(F1smoke)/Sum(f1pnlwt*flagF1smoke)) as F1smokeweight,

f1pnlwt * (count(F1drink)/Sum(f1pnlwt*flagF1drink)) as F1drinkweight,

f1pnlwt * (count(F1cutclass)/Sum(f1pnlwt*flagF1cutclass)) as F1cutclassweight,

f1pnlwt * (count(F1problem)/Sum(f1pnlwt*flagF1problem)) as F1problemweight,

f1pnlwt * (count(F1marijuana)/Sum(f1pnlwt*flagF1marijuana)) as F1marijuanaweight

from here.all;
· The use of Macro (also described in another one of my SAS tutorial)

This example executes PROC MIXED three times, each time with different outcome variable, in one run.  

%macro john (var1=);

proc mixed data=here.esm covtest noclprint;

title “Modeling &var1”;

weight precision_weight;

class IDclass IDstudent;

model &var1= /solution ddfm=kr;

random intercept /sub=IDstudent(IDclass);
/*level2*/

random intercept /sub=IDclass ; 

/*level3*/

run;

%mend john;

%john (var1=engagement);

%john (var1=another_variable2);

%john (var1=another_variable3);

 APPENDIX ODS: How to find available tables from SAS’s PROCEDURES
You can find out what tables are available, do this to any SAS procedure:

ods trace on;

SAS procedure here

ods trace off;

The log file will tell you what kind of tables are available.  

Available tables from PROC MIXED:

Output Added:

-------------

Name:       ModelInfo

Label:      Model Information

Template:   Stat.Mixed.ModelInfo

Path:       Mixed.ModelInfo

-------------

WARNING: Length of CLASS variable IDclass truncated to 16.

Output Added:

-------------

Name:       Dimensions

Label:      Dimensions

Template:   Stat.Mixed.Dimensions

Path:       Mixed.Dimensions

-------------

NOTE: 16 observations are not included because of missing values.

Output Added:

-------------

Name:       IterHistory

Label:      Iteration History

Template:   Stat.Mixed.IterHistory

Path:       Mixed.IterHistory

-------------

Output Added:

-------------

Name:       ConvergenceStatus

Label:      Convergence Status

Template:   Stat.Mixed.ConvergenceStatus

Path:       Mixed.ConvergenceStatus

-------------

NOTE: Convergence criteria met.

Output Added:

-------------

Name:       CovParms

Label:      Covariance Parameter Estimates

Template:   Stat.Mixed.CovParms

Path:       Mixed.CovParms

-------------

Output Added:

-------------

Name:       FitStatistics

Label:      Fit Statistics

Template:   Stat.Mixed.FitStatistics

Path:       Mixed.FitStatistics

-------------

Output Added:

-------------

Name:       SolutionF

Label:      Solution for Fixed Effects

Template:   Stat.Mixed.SolutionF

Path:       Mixed.SolutionF

-------------
VIII. Appendix Other resources for running PROC MIXED 

My TOP List
These are really good.  These also served as good references when I had to figure out how to do econometrics modeling using PROC MIXED.

· SAS System for Mixed Models By: Dr. Ramon C. Littell, Ph.D., Dr. George A. Milliken, Ph.D., Dr. Walter W. Stroup, Ph.D., and Dr. Russell D. Wolfinger, Ph.D.
· Applied Longitudinal Data Analysis: Modeling Change and Event Occurance.  Judith D. Singer and John B. Willett
· Hierarchical Linear Models: Applications and Data Analysis Methods.  Stephen W. Raudenbush and Anthony S. Bryk.
· Mixed Models Analyses Using the SAS System: Instructor-based training.  By SAS.
This one helped me understand how HLM and Rasch model can be related.

· Kamata, A. (2001). Item analysis by the hierarchical generalized linear model. Journal of Educational Measurement, 38, 79-93
I have not read this yet, but my friend Yasuo Miyazaki says Chapter 1 of this could be helpful for understanding how HLM and SEM are related.
· Multilevel Modeling of Educational Data.  Edited by:Ann A. O'Connell and D. Betsy McCoach.
General Reading 

· Multilevel stat book (all chapters on-line) by Goldstein http://www.arnoldpublishers.com/support/goldstein.htm
Gives a quick understanding when going from HLM to PROC MIXED

· Using SAS PROC MIXED to Fit Multilevel Models, Hierarchical Models, and Individual Growth Models, Judith D. Singer (Harvard University) Journal of Educational and Behavioral Statistics winter 1998, volume 23, number 4. A reprint of the paper downloadable at http://gseweb.harvard.edu/~faculty/singer/
· Suzuki and Sheu’s tutorial (PDF format) http://www.kcasug.org/mwsug1999/pdf/paper23.pdf
Comparison of HLM and PROC MIXED

· "Comparison of PROC MIXED in SAS and HLM for Hierarchical Linear Models" by Annie Qu says they produce almost identical results. http://www.pop.psu.edu/stat-core/software/hlm.htm 

Further details of PROC MIXED

· On-line SAS CD has a chapter on PROC MIXED.

Best place to go for technical details

· email support@sas.com with your SAS license registration information (Copy the head of the log file into the email text)

Mailing list for multilevel models. http://www.jiscmail.ac.uk/lists/multilevel.html
Nonlinear Model

· Dr. Wonfinger’s paper on NLMIXED procedure.  http://www.sas.com/rnd/app/papers/nlmixedsugi.pdf
· GLMMIX macro download at http://ewe3.sas.com/techsup/download/stat/glmm800.html
· How to use SAS for logistic Regression with Correlated Data by Kuss.  Some description of GMIMMIX and NLMIX.  http://www2.sas.com/proceedings/sugi27/p261-27.pdf
Applications done by friends/colleauges/myself
· Bidwell, Charles E., and Jefferey Y. Yasumoto.  1999.  “The Collegial Focus: Teaching Fields, Collegial Relationships, and Instructional Practice in American High Schools.”  Sociology of Education 72:234-56 (Social influence among high school teachers was modeled.  B&R’s HLM.)

· Yasumoto, Jeffrey Y., Kazuaki Uekawa, and Charles E. Bidwell.  2001.  “The Collegial Focus and High School Students’ Achievement.”  Sociology of Education 74: 181-209.  (Growth of students’ mathematics achievement score was modeled.  B&R’s HLM.)

· McFarland, Daniel A., 2001.  “Student Resistance: How the Formal and Informal Organization of Classrooms Facilitate Everyday Forms of Student Defiance.”  American Journal of Sociology Volume 107 Number 3 (November 2001): 612-78.  (Modeled the occurrence of students’ resistance in classrooms.  Multilevel Poisson Regression, SAS macro GLIMMIX.) 

· Uekawa, K., Borman K. M., Lee, R. (2007)  Student Engagement in America’s Urban High School Mathematics and Science Classrooms: Findings on Social Organization, Race, and Ethnicity.  The Urban Review, 39 (1), 1-106.

                                      The Mixed Procedure





                                       Model Information





                     Data Set                     HERE.ESM


                     Dependent Variable         engagement


                     Weight Variable             precision_weight


                     Covariance Structure       Variance Components


                     Subject Effects             IDstudent(IDclass),


                                                    IDclass


                     Estimation Method           REML


                     Residual Variance Method     Profile


                     Fixed Effects SE Method      Prasad-Rao-Jeske-


                                                  Kackar-Harville


                     Degrees of Freedom Method    Kenward-Roger








Dimensions





                              Covariance Parameters             3


                              Columns in X                        1


                              Columns in Z Per Subject         17


                              Subjects                          32


                              Max Obs Per Subject             114


                              Observations Used              2316


                              Observations Not Used             0


                              Total Observations             2316





                                     Non conditional model                                     4

                                      The Mixed Procedure





                                 Covariance Parameter Estimates





                                                         Standard         Z


         Cov Parm      Subject               Estimate       Error     Value        Pr Z





         Intercept     IDstudent(IDclass)   23.3556      2.5061      9.32      <.0001


         Intercept     IDclass                 5.3168      2.0947      2.54      0.0056


         Residual                               31.9932      1.0282     31.12      <.0001





Iteration History





                  Iteration    Evaluations    -2 Res Log Like       Criterion





                          0              1     16058.34056441


                          1              2     15484.50136495      0.00180629


                          2              1     15472.31142352      0.00029511


                          3              1     15470.46506873      0.00001303


                          4              1     15470.38907630      0.00000007


                          5              1     15470.38870052      0.00000000








                                   Convergence criteria met.











Fit Statistics





                             -2 Res Log Likelihood         15470.4


                             AIC (smaller is better)       15476.4


                             AICC (smaller is better)      15476.4


                             BIC (smaller is better)       15480.8





Solution for Fixed Effects





                                         Standard


               Effect        Estimate       Error      DF    t Value    Pr > |t|





               Intercept      -1.2315      0.5081    30.2      -2.42      0.0216





Insert all the higher level equations into level-1 equation.


Identify the variables that appear with error term.  For example, X appears with R1 (see R1*X).


Put that variable at a random statement.

